/*-----------------------------------------------------------------------------
	The contents of this file are subject to the Mozilla Public License
	Version 1.1 (the "License"); you may not use this file except in
	compliance with the License. You may obtain a copy of the License at
	http://www.mozilla.org/MPL/
	
	Software distributed under the License is distributed on an "AS IS"
	basis, WITHOUT WARRANTY OF ANY KIND, either express or implied. See the
	License for the specific language governing rights and limitations
	under the License.
	
	The Original Code is Fever Framework code.
	
	The Initial Developer of the Original Code is Romain Ecarnot.
	Portions created by Initial Developer are Copyright (C) 2006
	the Initial Developer. All Rights Reserved.
	
	Contributor(s):
	
	Unless required by applicable law or agreed to in writing, software
	distributed under the License is distributed on an "AS IS" BASIS,
	WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
	See the License for the specific language governing permissions and
	limitations under the License.
-----------------------------------------------------------------------------*/

import com.bourre.events.StringEvent;

import fever.app.browser.analytic.AnalyticStrategy;
import fever.commands.DelayCommandManager;
import fever.utils.Stringifier;

/**
 * Application analytic tools manager.
 * 
 * <p>Adds analytic tool using {@link #register()} method.
 * <ul>
 *   <li>{@link fever.app.browser.analytic.strategy.GoogleAnalytic</li> *   <li>{@link fever.app.browser.analytic.strategy.BlueStreak</li>
 *   <li>{@link fever.app.browser.analytic.strategy.Omniture</li>
 * </ul>
 * 
 * {@code
 *   var bsTool : BlueStreak = new BlueStreak();
 *   bsTool.baseURL = 'http://s0b.bluestreak.com/ix.e?bb&t=';
 *   
 *   var manager : AnalyticManager = AnalyticManager.getInstance();
 *   manager.register( new GoogleAnalytic() );
 *   manager.register( new Omniture() );
 *   manager.register( bsTool );
 * }
 * 
 * {@code
 *   AnalyticManager.getInstance().track( '/portfolio/recent/video/' );
 * }
 *
 * @author Romain Ecarnot
 */
class fever.app.browser.analytic.AnalyticManager
{
	//-------------------------------------------------------------------------
	// Private properties
	//-------------------------------------------------------------------------
	
	private static var _instance : AnalyticManager;
	private var _aTrackers : DelayCommandManager;
	
	
	//-------------------------------------------------------------------------
	// Public API
	//-------------------------------------------------------------------------
	
	/**
	 * Returns {@link AnalyticManager} instance.
	 */
	public static function getInstance() : AnalyticManager
	{
		if( !_instance ) _instance = new AnalyticManager();
		return _instance;	
	}
	
	/**
	 * 
	 */
	public function register( tracker : AnalyticStrategy ) : Void
	{
		_aTrackers.addCommand( tracker );
	}
	
	/**
	 * 
	 */
	public function unregister( tracker : AnalyticStrategy ) : Void
	{
		_aTrackers.removeCommand( tracker );
	}
	
	/**
	 * Track passed-in {@code link} page.
	 */
	public function track( link : String ) : Void
	{
		_aTrackers.execute( new StringEvent( null, link ) );
	}
	
	/**
	 * Returns string representation.
	 */
	public function toString() : String
	{
		return Stringifier.parse( this );
	}
	
	
	//-------------------------------------------------------------------------
	// Private implementation
	//-------------------------------------------------------------------------
	
	/**
	 * Constructor.
	 */
	private function AnalyticManager()
	{
		_aTrackers = new DelayCommandManager();
	}	
}